
Filtering
    

Mix.install([
  {:nx_signal, "~> 0.2"},
  {:vega_lite, "~> 0.1"},
  {:kino_vega_lite, "~> 0.1"}
])
Prepare the data
fs = 16.0e3
window_duration_seconds = 100.0e-3
window_length = 2 ** ceil(:math.log2(fs * window_duration_seconds))
signal_duration_seconds = 3
signal_length = ceil(signal_duration_seconds * fs)

sin = fn freq, n ->
  Nx.sin(Nx.multiply(2 * :math.pi() * freq / fs, n))
end

half_n = Nx.iota({div(signal_length, 2)})

sin220 = sin.(220, half_n)
sin440 = sin.(440, half_n)
sin1000 = sin.(440 * 5 / 2, half_n)
sin3000 = sin.(220 * 4 / 3 * 4, half_n)

n = Nx.iota({signal_length})
t = Nx.divide(n, fs)
data = Nx.concatenate([Nx.add(sin440, sin1000), Nx.add(sin220, sin3000)])

# Data for plotting

slice = (signal_length - 3000)..(signal_length - 2750)
plot_data = %{y: Nx.to_flat_list(data[[slice]]), x: Nx.to_flat_list(t[[slice]])}
VegaLite.new(width: 600, height: 400, title: "Signal sample")
|> VegaLite.data_from_values(plot_data, only: ["x", "y"])
|> VegaLite.mark(:line)
|> VegaLite.encode_field(:x, "x", type: :quantitative)
|> VegaLite.encode_field(:y, "y", type: :quantitative)
Preparing the Filter
# The filter used is a simple ideal filter specified through sinc

# Cutoff frequency in Hz
fc = 600
filter_indices = Nx.iota({window_length}) |> Nx.subtract(div(window_length, 2))

h_ideal =
  Nx.multiply(2 * fc / fs, NxSignal.Filters.sinc(Nx.multiply(2 * fc / fs, filter_indices)))

window = NxSignal.Windows.hann(n: window_length)

h = Nx.multiply(h_ideal, window)

hfft =
  h
  |> Nx.fft(length: window_length)
  |> Nx.abs()
  |> Nx.add(1.0e-10)

hfft_power =
  hfft
  |> Nx.log()
  |> Nx.divide(Nx.log(10))
  |> Nx.multiply(20)

f_idx = Nx.iota({window_length}) |> Nx.subtract(div(window_length, 2))

f = Nx.multiply(f_idx, fs / window_length)

plot_data = %{
  n: Enum.to_list(0..(window_length - 1)),
  h: Nx.to_flat_list(h),
  hfft:
    Nx.to_flat_list(
      Nx.take_along_axis(
        hfft_power,
        Nx.select(Nx.less(f_idx, 0), Nx.add(f_idx, window_length), f_idx)
      )
    ),
  f: Nx.to_flat_list(f)
}
VegaLite.new(
  width: 600,
  height: 400,
  title: "600 Hz Low-Pass filter (Hann window); L = 2048"
)
|> VegaLite.data_from_values(plot_data, only: ["n", "h"])
|> VegaLite.mark(:line)
|> VegaLite.encode_field(:x, "n", type: :quantitative)
|> VegaLite.encode_field(:y, "h", type: :quantitative)
VegaLite.new(
  width: 600,
  height: 400,
  title: "FFT - 1200 Hz Low-Pass filter (Hann window); L = 1764"
)
|> VegaLite.data_from_values(plot_data, only: ["f", "hfft"])
|> VegaLite.mark(:line)
|> VegaLite.encode_field(:x, "f", type: :quantitative)
|> VegaLite.encode_field(:y, "hfft", type: :quantitative)
Filtering the data
# Now that we have our filter, instead of convolving the filter
# with the time representation of the signal, we can multiply each
# STFT frame by the DFT of the filter (represented by hfft)

{z, t, f} =
  NxSignal.stft(data, window, fft_length: window_length, sampling_rate: fs, scaling: :spectrum)
# Filter
z_filtered = Nx.multiply(z, hfft)

max_f =
  Nx.select(Nx.greater_equal(f, fs / 2), Nx.iota(f.shape), Nx.size(f) + 1)
  |> Nx.argmin()
  |> Nx.to_number()

spectrogram = z |> Nx.slice([0, 0], [Nx.size(t), max_f]) |> Nx.abs() |> Nx.pow(2)

filtered_spectrogram =
  z_filtered |> Nx.slice([0, 0], [Nx.size(t), max_f]) |> Nx.abs() |> Nx.pow(2)

# Reconstruct the time signal
data_out =
  z_filtered
  |> NxSignal.istft(window, fft_length: window_length, scaling: :spectrum, sampling_rate: fs)
  |> Nx.as_type(data.type)
plot_data =
  for t_idx <- 0..(Nx.size(t) - 1),
      f_idx <- 0..max_f,
      Nx.to_number(f[[f_idx]]) <= 4000,
      reduce: %{"t" => [], "f" => [], "s" => [], "filtered_s" => []} do
    %{"t" => t_acc, "f" => f_acc, "s" => s_acc, "filtered_s" => filtered_s_acc} ->
      %{
        "t" => [Nx.to_number(t[[t_idx]]) | t_acc],
        "f" => [Float.round(Nx.to_number(f[[f_idx]]), 3) | f_acc],
        "s" => [Nx.to_number(spectrogram[[t_idx, f_idx]]) | s_acc],
        "filtered_s" => [Nx.to_number(filtered_spectrogram[[t_idx, f_idx]]) | filtered_s_acc]
      }
  end
defmodule Spectrogram do
  alias VegaLite, as: Vl

  def plot(title, dataset) do
    Vl.new(title: title, width: 500, height: 500)
    |> Vl.mark(:rect)
    |> Vl.data_from_values(dataset)
    |> Vl.encode_field(:x, "t",
      type: :quantitative,
      title: "Time (seconds)",
      axis: [tick_min_step: 0.1],
      grid: false
    )
    |> Vl.encode_field(:y, "f",
      type: :quantitative,
      sort: "-x",
      title: "Frequency (Hz)",
      axis: [tick_count: 25],
      grid: false
    )
    |> Vl.encode_field(:color, "s",
      aggregate: :max,
      type: :quantitative,
      scale: [scheme: "viridis"],
      legend: [title: "dBFS"]
    )
    |> Vl.config(view: [stroke: nil])
  end
end
Spectrogram.plot("Spectrogram", Map.take(plot_data, ["t", "f", "s"]))
Spectrogram.plot(
  "Filtered Spectrogram",
  Map.take(plot_data, ["t", "f"]) |> Map.put("s", plot_data["filtered_s"])
)
plot_data = %{
  y: Nx.to_flat_list(data_out[[slice]]),
  x: Nx.to_flat_list(Nx.divide(n, fs)[[slice]])
}
VegaLite.new(width: 600, height: 400, title: "Reconstructed Signal")
|> VegaLite.data_from_values(plot_data, only: ["x", "y"])
|> VegaLite.mark(:line)
|> VegaLite.encode_field(:x, "x", type: :quantitative)
|> VegaLite.encode_field(:y, "y", type: :quantitative)
As we can see, the same original slice is now a pure 220Hz sine wave, because we got rid of the higher frequency harmonic.



  

    
Spectrogram Plotting
    

Mix.install([
  {:nx_signal, "~> 0.2"},
  {:vega_lite, "~> 0.1.4"},
  {:kino_vega_lite, "~> 0.1.1"}
])
Generating the audio data
# You can load an audio file here.  For this example,
# we're producing 3 seconds of 220Hz, 440Hz, 1kHz and 3kHz sine waves

fs = 44.1e3
t_max = 3

full_n = ceil(fs * t_max)
half_n = div(full_n, 2)

# samples/sec * sec = samples
n = Nx.iota({half_n})

sin = fn freq, n ->
  Nx.sin(Nx.multiply(2 * :math.pi() * freq / fs, n))
end

sin220 = sin.(220, n)
sin440 = sin.(440, n)
sin1000 = sin.(1000, n)
sin3000 = sin.(3000, n)

data = Nx.concatenate([Nx.add(sin440, sin1000), Nx.add(sin220, sin3000)])
n = Nx.iota({full_n})

d = %{data: Nx.to_flat_list(data[[1000..1250]]), n: Nx.to_flat_list(n[[1000..1250]])}
VegaLite.new(width: 600, height: 600, title: "Audio Sample")
|> VegaLite.data_from_values(d, only: ["n", "data"])
|> VegaLite.mark(:line)
|> VegaLite.encode_field(:x, "n", type: :ordinal)
|> VegaLite.encode_field(:y, "data", type: :quantitative)
defmodule Spectrogram do
  alias VegaLite, as: Vl
  import Nx.Defn

  def calculate_stft_and_plot_spectrogram(
        input,
        fs,
        window_duration_ms,
        plot_cutoff_frequency \\ 4000
      ) do
    n_window = ceil(fs * window_duration_ms)
    {spectrogram, f, t, max_f} = stft(input, fs: fs, n_window: n_window)

    max_f = Nx.to_number(max_f)
    spectrogram = Nx.slice(spectrogram, [0, 0], [Nx.size(t), max_f])
    f = Nx.slice(f, [0], [max_f])

    spectrogram
    |> to_plot_data(f, t, plot_cutoff_frequency)
    |> plot()
  end

  defn stft(input, opts) do
    fs = opts[:fs]
    n_window = opts[:n_window]

    # ms to samples
    window = NxSignal.Windows.hann(n: n_window, is_periodic: true)

    # use the default overlap of 50%
    {s, t, f} = NxSignal.stft(input, window, sampling_rate: fs, fft_length: 1024)

    max_f =
      Nx.select(f >= fs / 2, Nx.iota(f.shape), Nx.size(f) + 1)
      |> Nx.argmin()

    spectrogram = Nx.abs(s)
    # to dBFS
    spectrogram = 20 * Nx.log(spectrogram / Nx.reduce_max(spectrogram)) / Nx.log(10)

    {spectrogram, f, t, max_f}
  end

  defp to_plot_data(s, f, t, plot_cutoff_frequency) do
    for t_idx <- 0..(Nx.size(t) - 1),
        f_idx <- 0..(Nx.size(f) - 1),
        Nx.to_number(f[[f_idx]]) <= plot_cutoff_frequency,
        reduce: %{"t" => [], "f" => [], "s" => []} do
      %{"t" => t_acc, "f" => f_acc, "s" => s_acc} ->
        %{
          "t" => [Nx.to_number(t[[t_idx]]) | t_acc],
          "f" => [Float.round(Nx.to_number(f[[f_idx]]), 3) | f_acc],
          "s" => [Nx.to_number(s[[t_idx, f_idx]]) | s_acc]
        }
    end
  end

  defp plot(dataset) do
    Vl.new(title: "Spectrogram", width: 500, height: 500)
    |> Vl.mark(:rect)
    |> Vl.data_from_values(dataset)
    |> Vl.encode_field(:x, "t",
      type: :quantitative,
      title: "Time (seconds)",
      axis: [tick_min_step: 0.1],
      grid: false
    )
    |> Vl.encode_field(:y, "f",
      type: :quantitative,
      sort: "-x",
      title: "Frequency (Hz)",
      axis: [tick_count: 25],
      grid: false
    )
    |> Vl.encode_field(:color, "s",
      aggregate: :max,
      type: :quantitative,
      scale: [scheme: "viridis"],
      legend: [title: "dBFS"]
    )
    |> Vl.config(view: [stroke: nil])
  end
end
Spectrogram.calculate_stft_and_plot_spectrogram(data, fs, 50.0e-3)
Notice how the first half of the spectrogram looks cleaner than the second one. This is due to the window length, that also interferes in how we can observe both our time and frequency resolutions.
Below we can see what happens if we use different window durations (150ms, 100ms and 25ms respectively).
Spectrogram.calculate_stft_and_plot_spectrogram(data, fs, 150.0e-3)
Spectrogram.calculate_stft_and_plot_spectrogram(data, fs, 100.0e-3)
Spectrogram.calculate_stft_and_plot_spectrogram(data, fs, 25.0e-3)

    Functions: Time-Frequency
  


    
      
        fft_frequencies(sampling_rate, opts \\ [])

      


        Computes the frequency bins for a FFT with given options.



    


    
      
        istft(data, window, opts)

      


        Computes the Inverse Short-Time Fourier Transform of a tensor.



    


    
      
        mel_filters(fft_length, mel_bins, sampling_rate, opts \\ [])

      


        Generates weights for converting an STFT representation into MEL-scale.



    


    
      
        stft(data, window, opts \\ [])

      


        Computes the Short-Time Fourier Transform of a tensor.



    


    
      
        stft_to_mel(z, sampling_rate, opts \\ [])

      


        Converts a given STFT time-frequency spectrum into a MEL-scale time-frequency spectrum.

        as_windowed(tensor, opts \\ [])

      


        Returns a tensor of K windows of length N



    


    
      
        overlap_and_add(tensor, opts \\ [])

      


        Performs the overlap-and-add algorithm over
an {..., M, N}-shaped tensor, where M is the number of
windows and N is the window size.

      Link to this function
    
    fft_frequencies(sampling_rate, opts \\ [])


      
       
       View Source
     


  


  

Computes the frequency bins for a FFT with given options.

  
  arguments

  
  Arguments


	sampling_rate - Sampling frequency in Hz.


  
  options

  
  Options


	:fft_length - Number of FFT frequency bins.
	:type - Optional output type. Defaults to {:f, 32}
	:name - Optional axis name for the tensor. Defaults to :frequencies


  
  examples

  
  Examples


iex> NxSignal.fft_frequencies(1.6e4, fft_length: 10)
#Nx.Tensor<
  f32[frequencies: 10]
  [0.0, 1.6e3, 3.2e3, 4.8e3, 6.4e3, 8.0e3, 9.6e3, 1.12e4, 1.28e4, 1.44e4]
>

  



  
    
      
      Link to this function
    
    istft(data, window, opts)


      
       
       View Source
     


  


  

Computes the Inverse Short-Time Fourier Transform of a tensor.
Returns a tensor of M time-domain frames of length fft_length.
See also: NxSignal.Windows, stft/3

  
  options

  
  Options


	:fft_length - the DFT length that will be passed to Nx.fft/2. Defaults to :power_of_two.
	:overlap_length - the number of samples for the overlap between frames.
Defaults to half the window size.
	:sampling_rate - the sampling rate $F_s$ in Hz. Defaults to 1000.
	:scaling - nil, :spectrum or :psd.	:spectrum - each frame is multiplied by $\sum_{i} window[i]$.
	nil - No scaling is applied.
	:psd - each frame is multiplied by $\sqrt{F_s\sum_{i} window[i]^2}$.




  
  examples

  
  Examples


In general, istft/3 takes in the same parameters and window as the stft/3 that generated the spectrum.
In the first example, we can notice that the reconstruction is mostly perfect, aside from the first sample.
This is because the Hann window only ensures perfect reconstruction in overlapping regions, so the edges
of the signal end up being distorted.
iex> t = Nx.tensor([10, 10, 1, 0, 10, 10, 2, 20])
iex> w = NxSignal.Windows.hann(n: 4)
iex> opts = [sampling_rate: 1, fft_length: 4]
iex> {z, _time, _freqs} = NxSignal.stft(t, w, opts)
iex> result = NxSignal.istft(z, w, opts)
iex> Nx.as_type(result, Nx.type(t))
#Nx.Tensor<
  s64[8]
  [0, 10, 1, 0, 10, 10, 2, 20]
>
Different scaling options are available (see stft/3 for a more detailed explanation).
For perfect reconstruction, you want to use the same scaling as the STFT:
iex> t = Nx.tensor([10, 10, 1, 0, 10, 10, 2, 20])
iex> w = NxSignal.Windows.hann(n: 4)
iex> opts = [scaling: :spectrum, sampling_rate: 1, fft_length: 4]
iex> {z, _time, _freqs} = NxSignal.stft(t, w, opts)
iex> result = NxSignal.istft(z, w, opts)
iex> Nx.as_type(result, Nx.type(t))
#Nx.Tensor<
  s64[8]
  [0, 10, 1, 0, 10, 10, 2, 20]
>

iex> t = Nx.tensor([10, 10, 1, 0, 10, 10, 2, 20], type: :f32)
iex> w = NxSignal.Windows.hann(n: 4)
iex> opts = [scaling: :psd, sampling_rate: 1, fft_length: 4]
iex> {z, _time, _freqs} = NxSignal.stft(t, w, opts)
iex> result = NxSignal.istft(z, w, opts)
iex> Nx.as_type(result, Nx.type(t))
#Nx.Tensor<
  f32[8]
  [0.0, 10.0, 0.9999999403953552, -2.1900146407460852e-7, 10.0, 10.0, 2.000000238418579, 20.0]
>

  



    

  
    
      
      Link to this function
    
    mel_filters(fft_length, mel_bins, sampling_rate, opts \\ [])


      
       
       View Source
     


  


  

Generates weights for converting an STFT representation into MEL-scale.
See also: stft/3, istft/3, stft_to_mel/3

  
  arguments

  
  Arguments


	fft_length - Number of FFT bins
	mel_bins - Number of target MEL bins
	sampling_rate - Sampling frequency in Hz


  
  options

  
  Options


	:max_mel - the pitch for the last MEL bin before log scaling. Defaults to 3016
	:mel_frequency_spacing - the distance in Hz between two MEL bins before log scaling. Defaults to 66.6
	:type - Target output type. Defaults to {:f, 32}


  
  examples

  
  Examples


iex> NxSignal.mel_filters(10, 5, 8.0e3)
#Nx.Tensor<
  f32[mels: 5][frequencies: 10]
  [
    [0.0, 8.129207999445498e-4, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0],
    [0.0, 9.972016559913754e-4, 2.1870288765057921e-4, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0],
    [0.0, 0.0, 9.510891977697611e-4, 4.150509194005281e-4, 0.0, 0.0, 0.0, 0.0, 0.0, 0.0],
    [0.0, 0.0, 0.0, 4.035891906823963e-4, 5.276656011119485e-4, 2.574124082457274e-4, 0.0, 0.0, 0.0, 0.0],
    [0.0, 0.0, 0.0, 0.0, 7.329034269787371e-5, 2.342205698369071e-4, 3.8295105332508683e-4, 2.8712040511891246e-4, 1.9128978601656854e-4, 9.545915963826701e-5]
  ]
>

  



    

  
    
      
      Link to this function
    
    stft(data, window, opts \\ [])


      
       
       View Source
     


  


  

Computes the Short-Time Fourier Transform of a tensor.
Returns the complex spectrum Z, the time in seconds for
each frame and the frequency bins in Hz.
The STFT is parameterized through:
	$k$: length of the Discrete Fourier Transform (DFT)
	$N$: length of each frame
	$H$: hop (in samples) between frames (calculated as $H = N - \text{overlap\_length}$)
	$M$: number of frames
	$x[n]$: the input time-domain signal
	$w[n]$: the window function to be applied to each frame

$$
DFT(x, w) := \sum_{n=0}^{N - 1} x[n]w[n]e^\frac{-2 \pi i k n}{N} \\
X[m, k] = DFT(x[mH..(mH + N - 1)], w)
$$
where $m$ assumes all values in the interval $[0, M - 1]$
See also: NxSignal.Windows, istft/3, stft_to_mel/3

  
  options

  
  Options


	:sampling_rate - the sampling frequency $F_s$ for the input in Hz. Defaults to 1000.
	:fft_length - the DFT length that will be passed to Nx.fft/2. Defaults to :power_of_two.
	:overlap_length - the number of samples for the overlap between frames.
Defaults to half the window size.
	:window_padding - :reflect, :zeros or nil. See as_windowed/3 for more details.
	:scaling - nil, :spectrum or :psd.	:spectrum - each frame is divided by $\sum_{i} window[i]$.
	nil - No scaling is applied.
	:psd - each frame is divided by $\sqrt{F_s\sum_{i} window[i]^2}$.




  
  examples

  
  Examples


iex> {z, t, f} = NxSignal.stft(Nx.iota({4}), NxSignal.Windows.rectangular(n: 2), overlap_length: 1, fft_length: 2, sampling_rate: 400)
iex> z
#Nx.Tensor<
  c64[frames: 3][frequencies: 2]
  [
    [1.0+0.0i, -1.0+0.0i],
    [3.0+0.0i, -1.0+0.0i],
    [5.0+0.0i, -1.0+0.0i]
  ]
>
iex> t
#Nx.Tensor<
  f32[frames: 3]
  [0.0024999999441206455, 0.004999999888241291, 0.007499999832361937]
>
iex> f
#Nx.Tensor<
  f32[frequencies: 2]
  [0.0, 200.0]
>

  



    

  
    
      
      Link to this function
    
    stft_to_mel(z, sampling_rate, opts \\ [])


      
       
       View Source
     


  


  

Converts a given STFT time-frequency spectrum into a MEL-scale time-frequency spectrum.
See also: stft/3, istft/3, mel_filters/4

  
  arguments

  
  Arguments


	z - STFT spectrum
	sampling_rate - Sampling frequency in Hz


  
  options

  
  Options


	:fft_length - Number of FFT bins
	:mel_bins - Number of target MEL bins. Defaults to 128
	:type - Target output type. Defaults to {:f, 32}


  
  examples

  
  Examples


iex> fft_length = 16
iex> sampling_rate = 8.0e3
iex> {z, _, _} = NxSignal.stft(Nx.iota({10}), NxSignal.Windows.hann(n: 4), overlap_length: 2, fft_length: fft_length, sampling_rate: sampling_rate, window_padding: :reflect)
iex> Nx.axis_size(z, :frequencies)
16
iex> Nx.axis_size(z, :frames)
6
iex> NxSignal.stft_to_mel(z, sampling_rate, fft_length: fft_length, mel_bins: 4)
#Nx.Tensor<
  f32[frames: 6][mel: 4]
  [
    [0.2900530695915222, 0.17422175407409668, 0.18422472476959229, 0.09807997941970825],
    [0.6093881130218506, 0.5647397041320801, 0.4353824257850647, 0.08635270595550537],
    [0.7584103345870972, 0.7085014581680298, 0.5636920928955078, 0.179118812084198],
    [0.8461772203445435, 0.7952491044998169, 0.6470762491226196, 0.2520409822463989],
    [0.908548891544342, 0.8572604656219482, 0.7078656554222107, 0.3086767792701721],
    [0.908548891544342, 0.8572604656219482, 0.7078656554222107, 0.3086767792701721]
  ]
>

      Link to this function
    
    as_windowed(tensor, opts \\ [])


      
       
       View Source
     


  


  

Returns a tensor of K windows of length N

  
  options

  
  Options


	:window_length - the number of samples in a window
	:stride - The number of samples to skip between windows. Defaults to 1.
	:padding - Padding mode, can be :reflect or a valid padding as per Nx.pad/3 over the
input tensor's shape. Defaults to :valid. If :reflect or :same, the first window will be centered
at the start of the signal. The padding is applied for the whole input, rather than individual
windows. For :zeros, effectively each incomplete window will be zero-padded.


  
  examples

  
  Examples


iex> NxSignal.as_windowed(Nx.tensor([0, 1, 2, 3, 4, 10, 11, 12]), window_length: 4)
#Nx.Tensor<
  s64[5][4]
  [
    [0, 1, 2, 3],
    [1, 2, 3, 4],
    [2, 3, 4, 10],
    [3, 4, 10, 11],
    [4, 10, 11, 12]
  ]
>

iex> NxSignal.as_windowed(Nx.tensor([0, 1, 2, 3, 4, 10, 11, 12]), window_length: 3)
#Nx.Tensor<
  s64[6][3]
  [
    [0, 1, 2],
    [1, 2, 3],
    [2, 3, 4],
    [3, 4, 10],
    [4, 10, 11],
    [10, 11, 12]
  ]
>

iex> NxSignal.as_windowed(Nx.tensor([0, 1, 2, 3, 4, 10, 11]), window_length: 2, stride: 2, padding: [{0, 3}])
#Nx.Tensor<
  s64[5][2]
  [
    [0, 1],
    [2, 3],
    [4, 10],
    [11, 0],
    [0, 0]
  ]
>

iex> t = Nx.iota({7});
iex> NxSignal.as_windowed(t, window_length: 6, padding: :reflect, stride: 1)
#Nx.Tensor<
  s64[8][6]
  [
    [3, 2, 1, 0, 1, 2],
    [2, 1, 0, 1, 2, 3],
    [1, 0, 1, 2, 3, 4],
    [0, 1, 2, 3, 4, 5],
    [1, 2, 3, 4, 5, 6],
    [2, 3, 4, 5, 6, 5],
    [3, 4, 5, 6, 5, 4],
    [4, 5, 6, 5, 4, 3]
  ]
>

iex> NxSignal.as_windowed(Nx.iota({10}), window_length: 6, padding: :reflect, stride: 2)
#Nx.Tensor<
  s64[6][6]
  [
    [3, 2, 1, 0, 1, 2],
    [1, 0, 1, 2, 3, 4],
    [1, 2, 3, 4, 5, 6],
    [3, 4, 5, 6, 7, 8],
    [5, 6, 7, 8, 9, 8],
    [7, 8, 9, 8, 7, 6]
  ]
>

  



    

  
    
      
      Link to this function
    
    overlap_and_add(tensor, opts \\ [])


      
       
       View Source
     


  


  

Performs the overlap-and-add algorithm over
an {..., M, N}-shaped tensor, where M is the number of
windows and N is the window size.
The tensor is zero-padded on the right so
the last window fully appears in the result.

  
  options

  
  Options


	:overlap_length - The number of overlapping samples between windows
	:type - output type for casting the accumulated result.
If not given, defaults to Nx.Type.to_complex/1 called on the input type.


  
  examples

  
  Examples


iex> NxSignal.overlap_and_add(Nx.iota({3, 4}), overlap_length: 0)
#Nx.Tensor<
  s64[12]
  [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11]
>

iex> NxSignal.overlap_and_add(Nx.iota({3, 4}), overlap_length: 3)
#Nx.Tensor<
  s64[6]
  [0, 5, 15, 18, 17, 11]
>

iex> t = Nx.tensor([[[[0, 1, 2, 3], [4, 5, 6, 7]]], [[[10, 11, 12, 13], [14, 15, 16, 17]]]]) |> Nx.vectorize(x: 2, y: 1)
iex> NxSignal.overlap_and_add(t, overlap_length: 3)
#Nx.Tensor<
  vectorized[x: 2][y: 1]
  s64[5]
  [
    [
      [0, 5, 7, 9, 7]
    ],
    [
      [10, 25, 27, 29, 17]
    ]
  ]
>

  


        

      



  

    
NxSignal.Filters 
    



      
Common filter functions.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions: Filters
  


    
      
        sinc(t)

      


        Calculates the normalized sinc function $sinc(t) = \frac{sin(\pi t)}{\pi t}$



    





      


      
        
          
            
            Anchor for this section
          
Functions: Filters
        

        


  
    
      
      Link to this function
    
    sinc(t)


      
       
       View Source
     


  


  

Calculates the normalized sinc function $sinc(t) = \frac{sin(\pi t)}{\pi t}$

  
  examples

  
  Examples


iex> NxSignal.Filters.sinc(Nx.tensor([0, 0.25, 1]))
#Nx.Tensor<
  f32[3]
  [1.0, 0.9003162980079651, -2.7827534054836178e-8]
>

  


        

      



  

    
NxSignal.PeakFinding 
    



      
Peak finding algorithms.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        argrelextrema(data, comparator_fn, opts \\ [])

      


        Finds a relative extrema along the selected :axis.



    


    
      
        argrelmax(data, opts \\ [])

      


        Finds a relative maximum along the selected :axis.



    


    
      
        argrelmin(data, opts \\ [])

      


        Finds a relative minimum along the selected :axis.



    





      


      
        
          
            
            Anchor for this section
          
Functions
        

        


    

  
    
      
      Link to this function
    
    argrelextrema(data, comparator_fn, opts \\ [])


      
       
       View Source
     


  


  

Finds a relative extrema along the selected :axis.
A relative extremum is defined by the given comparator_fn
function of arity 2 function that returns a boolean tensor.
This is the function upon which &argrelmax/2 and &argrelmin/2
are implemented.
Returns a map in the following format:
%{
  indices: #Nx.Tensor<...>,
  valid_indices: #Nx.Tensor<...>
}
	:indices - the {n, rank} tensor of indices.
Contains -1 as a placeholder for invalid indices.

	:valid_indices - the number of valid indices that lead the tensor.



  
  options

  
  Options


	:axis - the axis along which to do comparisons. Defaults to 0.
	:order - the number of neighbor samples considered for the
comparison in each direction. Defaults to 1.


  
  examples

  
  Examples


First, do read the examples on argrelmax/2 keeping in mind that
it is equivalent to argrelextrema(&1, &Nx.greater/2, &2), as well
as argrelmin/2 which is equivalent to argrelextrema(&1, &Nx.less/2, &2).
Having that in mind, we will expand on those concepts by using a custom function.
For instance, we can change the definition of a relative maximum to one where
a number is a relative maximum if it is greater than or equal to the double of its
neighbors, as follows:
iex> comparator = fn x, y -> Nx.greater_equal(x, Nx.multiply(y, 2)) end
iex> x = Nx.tensor([0, 1, 3, 2, 0, 1, 0, 0, 0, 2, 1])
iex> result = NxSignal.PeakFinding.argrelextrema(x, comparator)
iex> result.valid_indices
#Nx.Tensor<
  u64
  3
>
iex> result.indices[0..2]
#Nx.Tensor<
  s64[3][1]
  [
    [5],
    [7],
    [9]
  ]
>
Same applies for finding local minima. In the next example, we
find all local minima (i.e. &Nx.less/2) that are
different to the global minimum.
iex> x = Nx.tensor([0, 1, 0, 2, 1, 3, 0, 1])
iex> global_minimum = Nx.reduce_min(x)
iex> comparator = fn x, y ->
...> x_not_global = Nx.not_equal(x, global_minimum)
...> y_not_global = Nx.not_equal(y, global_minimum)
...> both_not_global = Nx.logical_and(x_not_global, y_not_global)
...> Nx.logical_and(Nx.less(x, y), both_not_global)
...> end
iex> result = NxSignal.PeakFinding.argrelextrema(x, comparator)
iex> result.valid_indices
#Nx.Tensor<
  u64
  1
>
iex> result.indices[0..0]
#Nx.Tensor<
  s64[1][1]
  [
    [4]
  ]
>

  



    

  
    
      
      Link to this function
    
    argrelmax(data, opts \\ [])


      
       
       View Source
     


  


  

Finds a relative maximum along the selected :axis.
A relative maximum is defined by the element being greater
than its neighbors along the axis :axis.
Returns a map in the following format:
%{
  indices: #Nx.Tensor<...>,
  valid_indices: #Nx.Tensor<...>
}
	:indices - the {n, rank} tensor of indices.
Contains -1 as a placeholder for invalid indices.

	:valid_indices - the number of valid indices that lead the tensor.



  
  options

  
  Options


	:axis - the axis along which to do comparisons. Defaults to 0.
	:order - the number of neighbor samples considered for the
comparison in each direction. Defaults to 1.


  
  examples

  
  Examples


iex> x = Nx.tensor([2, 1, 2, 3, 2, 0, 1, 0])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmax(x)
iex> valid_indices
#Nx.Tensor<
  u64
  2
>
iex> indices
#Nx.Tensor<
  s64[8][1]
  [
    [3],
    [6],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1]
  ]
>
iex> Nx.slice_along_axis(indices, 0, Nx.to_number(valid_indices), axis: 0)
#Nx.Tensor<
  s64[2][1]
  [
    [3],
    [6]
  ]
>
For the same tensor in the previous example, we can use :order to check if
the relative maxima are extrema in a wider neighborhood.
iex> x = Nx.tensor([2, 1, 2, 3, 2, 0, 1, 0])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmax(x, order: 3)
iex> valid_indices
#Nx.Tensor<
  u64
  1
>
iex> indices
#Nx.Tensor<
  s64[8][1]
  [
    [3],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1]
  ]
>
iex> Nx.slice_along_axis(indices, 0, Nx.to_number(valid_indices), axis: 0)
#Nx.Tensor<
  s64[1][1]
  [
    [3]
  ]
>
We can also apply this function to tensors with a larger rank:
iex> x = Nx.tensor([[1, 2, 1, 2], [6, 2, 0, 0], [5, 3, 4, 4]])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmax(x)
iex> valid_indices
#Nx.Tensor<
  u64
  1
>
iex> indices[0]
#Nx.Tensor<
  s64[2]
  [1, 0]
>
iex> %{indices: indices} = NxSignal.PeakFinding.argrelmax(x, axis: 1)
iex> valid_indices
#Nx.Tensor<
  u64
  1
>
iex> indices[0]
#Nx.Tensor<
  s64[2]
  [0, 1]
>

  



    

  
    
      
      Link to this function
    
    argrelmin(data, opts \\ [])


      
       
       View Source
     


  


  

Finds a relative minimum along the selected :axis.
A relative minimum is defined by the element being greater
than its neighbors along the axis :axis.
Returns a map in the following format:
%{
  indices: #Nx.Tensor<...>,
  valid_indices: #Nx.Tensor<...>
}
	:indices - the {n, rank} tensor of indices.
Contains -1 as a placeholder for invalid indices.

	:valid_indices - the number of valid indices that lead the tensor.



  
  options

  
  Options


	:axis - the axis along which to do comparisons. Defaults to 0.
	:order - the number of neighbor samples considered for the
comparison in each direction. Defaults to 1.


  
  examples

  
  Examples


iex> x = Nx.tensor([2, 1, 2, 3, 2, 0, 1, 0])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmin(x)
iex> valid_indices
#Nx.Tensor<
  u64
  2
>
iex> indices
#Nx.Tensor<
  s64[8][1]
  [
    [1],
    [5],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1]
  ]
>
iex> Nx.slice_along_axis(indices, 0, Nx.to_number(valid_indices), axis: 0)
#Nx.Tensor<
  s64[2][1]
  [
    [1],
    [5]
  ]
>
For the same tensor in the previous example, we can use :order to check if
the relative maxima are extrema in a wider neighborhood.
iex> x = Nx.tensor([2, 1, 2, 3, 2, 0, 1, 0])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmin(x, order: 3)
iex> valid_indices
#Nx.Tensor<
  u64
  1
>
iex> indices
#Nx.Tensor<
  s64[8][1]
  [
    [1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1],
    [-1]
  ]
>
iex> Nx.slice_along_axis(indices, 0, Nx.to_number(valid_indices), axis: 0)
#Nx.Tensor<
  s64[1][1]
  [
    [1]
  ]
>
We can also apply this function to tensors with a larger rank:
iex> x = Nx.tensor([[1, 2, 1, 2], [6, 2, 0, 0], [5, 3, 4, 4]])
iex> %{indices: indices, valid_indices: valid_indices} = NxSignal.PeakFinding.argrelmin(x)
iex> valid_indices
#Nx.Tensor<
  u64
  2
>
iex> indices[0..1]
#Nx.Tensor<
  s64[2][2]
  [
    [1, 2],
    [1, 3]
  ]
>
iex> %{indices: indices} = NxSignal.PeakFinding.argrelmin(x, axis: 1)
iex> valid_indices
#Nx.Tensor<
  u64
  2
>
iex> indices[0..1]
#Nx.Tensor<
  s64[2][2]
  [
    [0, 2],
    [2, 1]
  ]
>

  


        

      



  

    
NxSignal.Waveforms 
    



      
Functions that calculate waveforms given a time tensor.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions: Waveforms
  


    
      
        chirp(t, f0, t1, f1, opts \\ [])

      


        Chirp function.



    


    
      
        gaussian_pulse(t, opts \\ [])

      


        Gaussian modulated sinusoid.



    


    
      
        polynomial_sweep(t, coefs, opts \\ [])

      


        Frequency-swept cosine generator, with a time-dependent frequency.



    


    
      
        sawtooth(t, opts \\ [])

      


        Periodic sawtooth or triangular waveform.



    


    
      
        square(t, opts \\ [])

      


        A periodic square wave with period $2\pi$.



    


    
      
        unit_impulse(shape, opts \\ [])

      


        Discrete delta function or unit basis vector.



    





      


      
        
          
            
            Anchor for this section
          
Functions: Waveforms
        

        


    

  
    
      
      Link to this function
    
    chirp(t, f0, t1, f1, opts \\ [])


      
       
       View Source
     


  


  

Chirp function.
Starts at t with frequency f0 and ends at t1 with
frequency f1.

  
  options

  
  Options


	:phi - phase shift for the chirp.
	:vertex_zero - determines the position of the parabolic vertex
for when method: :quadratic. Defaults to true.
	:method - One of various frequency interpolation methods:	:linear - linear interpolation.
	:quadratic - parabolic interpolation with vertex at t1 or t0,
depending if vertex_zero: false or vertex_zero: true respectively.
	:hyperbolic - hyperbolic interpolation.
	:logarithmic - logarithmic (also known as geometric or exponential)
interpolation. f0 and f1 must be non-zero and have the same sign.




  
  examples

  
  Examples


iex> t = Nx.linspace(0, 10, n: 5)
iex> NxSignal.Waveforms.chirp(t, 10, 10, 1, method: :linear)
#Nx.Tensor<
  f32[5]
  [1.0, 0.38268470764160156, 3.795033308051643e-6, -0.382683128118515, 1.0]
>
iex> NxSignal.Waveforms.chirp(t, 10, 10, 1, method: :quadratic)
#Nx.Tensor<
  f32[5]
  [1.0, -0.9807833433151245, -9.958475288840418e-8, -0.5555803775787354, 1.0]
>
iex> NxSignal.Waveforms.chirp(t, 10, 10, 1, method: :quadratic, vertex_zero: false)
#Nx.Tensor<
  f32[5]
  [1.0, 0.5555850863456726, -7.490481493732659e-6, 0.98078453540802, 1.0]
>
iex> NxSignal.Waveforms.chirp(t, 10, 10, 1, method: :hyperbolic)
#Nx.Tensor<
  f32[5]
  [1.0, 0.8229323029518127, 0.9335360527038574, 0.013466471806168556, -0.8630329966545105]
>
iex> NxSignal.Waveforms.chirp(t, 10, 10, 1, method: :logarithmic)
#Nx.Tensor<
  f32[5]
  [1.0, 0.9989554286003113, -0.33371755480766296, -0.2700612545013428, 0.8558982610702515]
>

  



    

  
    
      
      Link to this function
    
    gaussian_pulse(t, opts \\ [])


      
       
       View Source
     


  


  

Gaussian modulated sinusoid.
The returned value follows the formula:
$$
f(t) = e^{-at^2}(cos(2 \pi f_c t) + isin(2 \pi f_c t))
$$
Where the exponential envelope is returned as envelope,
and the real and imaginary parts of $f(t)$ are returned as
in_phase and quadrature in the output map.
Note that in_phase and quadrature are are equivalent to
$\operatorname{Re} \lbrace f(t) \rbrace$ and $\operatorname{Im} \lbrace f(t) \rbrace$ respectively.

  
  examples

  
  Examples


iex> t = Nx.linspace(0, 1, n: 4)
iex> pulse = NxSignal.Waveforms.gaussian_pulse(t, center_frequency: 4)
iex> pulse.envelope
#Nx.Tensor<
  f32[4]
  [1.0, 0.20443114638328552, 0.001746579073369503, 6.236254534996988e-7]
>
iex> pulse.in_phase
#Nx.Tensor<
  f32[4]
  [1.0, -0.10221561044454575, -8.732887799851596e-4, 6.236254534996988e-7]
>
iex> pulse.quadrature
#Nx.Tensor<
  f32[4]
  [0.0, 0.17704254388809204, -0.0015125821810215712, 4.361525517918713e-13]
>

iex> t = Nx.linspace(0, 1, n: 4)
iex> pulse = NxSignal.Waveforms.gaussian_pulse(t, center_frequency: 4, bandwidth: 0.25)
iex> pulse.envelope
#Nx.Tensor<
  f32[4]
  [1.0, 0.6724140048027039, 0.20443114638328552, 0.028101593255996704]
>
iex> pulse.in_phase
#Nx.Tensor<
  f32[4]
  [1.0, -0.3362071216106415, -0.1022154912352562, 0.028101593255996704]
>
iex> pulse.quadrature
#Nx.Tensor<
  f32[4]
  [0.0, 0.5823275446891785, -0.177042618393898, 1.9653754179671523e-8]
>

  



    

  
    
      
      Link to this function
    
    polynomial_sweep(t, coefs, opts \\ [])


      
       
       View Source
     


  


  

Frequency-swept cosine generator, with a time-dependent frequency.
This function generates a sinusoidal function whose instantaneous
frequency varies with time. The frequency at time t is given by
the polynomial specified by the coefficients contained in coefs.
See also: chirp/5

  
  options

  
  Options


	:phi - phase shift to be applied before calculating the Nx.cos
for the output. Defaults to 0.
	:phi_unit - determines if :phi is given in :radians or :degrees.
Defaults to :radians.


  
  examples

  
  Examples


iex> t = Nx.linspace(0, 10, n: 5)
iex> NxSignal.Waveforms.polynomial_sweep(t, Nx.tensor([2, 0, 1]))
#Nx.Tensor<
  f32[5]
  [1.0, 0.866027295589447, -0.500006377696991, 1.7942518752533942e-5, -0.49998921155929565]
>
iex> NxSignal.Waveforms.polynomial_sweep(t, Nx.tensor([2, 0, 1]), phi: :math.pi() / 2)
#Nx.Tensor<
  f32[5]
  [-4.371138828673793e-8, 0.499999463558197, -0.8660194873809814, 1.0, 0.8660338521003723]
>
iex> NxSignal.Waveforms.polynomial_sweep(t, Nx.tensor([1, 0]))
#Nx.Tensor<
  f32[5]
  [1.0, 0.7071065306663513, -1.0, 0.7071084976196289, 1.0]
>
iex> NxSignal.Waveforms.polynomial_sweep(t, Nx.tensor([1, 0]), phi: 180, phi_unit: :degrees)
#Nx.Tensor<
  f32[5]
  [-1.0, -0.7071069478988647, 1.0, -0.7071129679679871, -1.0]
>

  



    

  
    
      
      Link to this function
    
    sawtooth(t, opts \\ [])


      
       
       View Source
     


  


  

Periodic sawtooth or triangular waveform.
The wave as a period of $2\pi$, rising from -1 to 1
in the interval $[0, 2\pi\cdot\text{width}]$ and dropping from
1 to -1 in the interval $[2\pi \cdot \text{width}, 2\pi]$.

  
  options

  
  Options


	:width - the width of the sawtooth. Must be a number
between 0 and 1 (both inclusive). Defaults to 1.


  
  examples

  
  Examples


A 5Hz waveform sampled at 500Hz for 1 second can be defined as:
t = Nx.linspace(0, 1, n: 500)
n = Nx.multiply(2 * :math.pi() * 5, t)
wave = NxSignal.Waveforms.sawtooth(n)

  



    

  
    
      
      Link to this function
    
    square(t, opts \\ [])


      
       
       View Source
     


  


  

A periodic square wave with period $2\pi$.
Evaluates to 1 in the interval $[0, 2\pi\text{duty}]$
and -1 in the interval $[2\pi\text{duty}, 2\pi]$.

  
  options

  
  Options


	:duty - a number or tensor representing the duty cycle.
If a tensor is given, the waveform changes over time, and it
must have the same length as the t input. Defaults to 0.5.


  
  examples

  
  Examples


iex> t = Nx.iota({10}) |> Nx.multiply(:math.pi() * 2 / 10)
iex> NxSignal.Waveforms.square(t, duty: 0.1)
#Nx.Tensor<
  s64[10]
  [1, -1, -1, -1, -1, -1, -1, -1, -1, -1]
>
iex> NxSignal.Waveforms.square(t, duty: 0.5)
#Nx.Tensor<
  s64[10]
  [1, 1, 1, 1, 1, -1, -1, -1, -1, -1]
>
iex> NxSignal.Waveforms.square(t, duty: 1)
#Nx.Tensor<
  s64[10]
  [1, 1, 1, 1, 1, 1, 1, 1, 1, 1]
>

iex> t = Nx.iota({10}) |> Nx.multiply(:math.pi() * 2 / 10)
iex> duty = Nx.tensor([0.1, 0, 0.3, 0, 0.5, 0, 0.7, 0, 0.9, 0])
iex> NxSignal.Waveforms.square(t, duty: duty)
#Nx.Tensor<
  s64[10]
  [1, -1, 1, -1, 1, -1, 1, -1, 1, -1]
>

  



    

  
    
      
      Link to this function
    
    unit_impulse(shape, opts \\ [])


      
       
       View Source
     


  


  

Discrete delta function or unit basis vector.

  
  options

  
  Options


	:index - one of number, numerical tensor
with length equal to the rank of the given
shape, or :midpoint. index: :midpoint,
is a shortcut for inserting the impulse
at the index which corresponds to half of
each dimension. Defaults to 0.

	:type - datatype for the output. Defaults to :f32.



  
  examples

  
  Examples


iex> NxSignal.Waveforms.unit_impulse({2})
#Nx.Tensor<
  f32[2]
  [1.0, 0.0]
>

iex> NxSignal.Waveforms.unit_impulse({3, 5}, type: :s64, index: :midpoint)
#Nx.Tensor<
  s64[3][5]
  [
    [0, 0, 0, 0, 0],
    [0, 0, 1, 0, 0],
    [0, 0, 0, 0, 0]
  ]
>

iex> NxSignal.Waveforms.unit_impulse({3, 5}, index: Nx.tensor([[2, 3]]), type: :s64)
#Nx.Tensor<
  s64[3][5]
  [
    [0, 0, 0, 0, 0],
    [0, 0, 0, 0, 0],
    [0, 0, 0, 1, 0]
  ]
>

  


        

      



  

    
NxSignal.Windows 
    



      
Common window functions.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions: Windowing
  


    
      
        bartlett(opts \\ [])

      


        Bartlett triangular window.



    


    
      
        blackman(opts \\ [])

      


        Blackman window.



    


    
      
        hamming(opts \\ [])

      


        Hamming window.



    


    
      
        hann(opts \\ [])

      


        Hann window.



    


    
      
        rectangular(opts \\ [])

      


        Rectangular window.



    


    
      
        triangular(opts \\ [])

      


        Triangular window.



    





      


      
        
          
            
            Anchor for this section
          
Functions: Windowing
        

        


    

  
    
      
      Link to this function
    
    bartlett(opts \\ [])


      
       
       View Source
     


  


  

Bartlett triangular window.
See also: triangular/1

  
  options

  
  Options


	:n - The window length. Mandatory option.
	:type - the output type for the window. Defaults to {:f, 32}
	:name - the axis name. Defaults to nil


  
  examples

  
  Examples


iex> NxSignal.Windows.bartlett(n: 3)
#Nx.Tensor<
  f32[3]
  [0.0, 0.6666666865348816, 0.6666666269302368]
>

  



    

  
    
      
      Link to this function
    
    blackman(opts \\ [])


      
       
       View Source
     


  


  

Blackman window.

  
  options

  
  Options


	:n - The window length. Mandatory option.
	:is_periodic - If true, produces a periodic window,
 otherwise produces a symmetric window. Defaults to true
	:type - the output type for the window. Defaults to {:f, 32}
	:name - the axis name. Defaults to nil


  
  examples

  
  Examples


iex> NxSignal.Windows.blackman(n: 5, is_periodic: false)
#Nx.Tensor<
  f32[5]
  [-1.4901161193847656e-8, 0.3400000333786011, 0.9999999403953552, 0.3400000333786011, -1.4901161193847656e-8]
>

iex> NxSignal.Windows.blackman(n: 5, is_periodic: true)
#Nx.Tensor<
  f32[5]
  [-1.4901161193847656e-8, 0.20077012479305267, 0.8492299318313599, 0.8492299318313599, 0.20077012479305267]
>

iex> NxSignal.Windows.blackman(n: 6, is_periodic: true, type: {:f, 32})
#Nx.Tensor<
  f32[6]
  [-1.4901161193847656e-8, 0.12999999523162842, 0.6299999952316284, 0.9999999403953552, 0.6299999952316284, 0.12999999523162842]
>

  



    

  
    
      
      Link to this function
    
    hamming(opts \\ [])


      
       
       View Source
     


  


  

Hamming window.

  
  options

  
  Options


	:n - The window length. Mandatory option.
	:is_periodic - If true, produces a periodic window,
 otherwise produces a symmetric window. Defaults to true
	:type - the output type for the window. Defaults to {:f, 32}
	:name - the axis name. Defaults to nil


  
  examples

  
  Examples


iex> NxSignal.Windows.hamming(n: 5, is_periodic: true)
#Nx.Tensor<
  f32[5]
  [0.08000001311302185, 0.39785221219062805, 0.9121478796005249, 0.9121478199958801, 0.3978521227836609]
>
iex> NxSignal.Windows.hamming(n: 5, is_periodic: false)
#Nx.Tensor<
  f32[5]
  [0.08000001311302185, 0.5400000214576721, 1.0, 0.5400000214576721, 0.08000001311302185]
>

  



    

  
    
      
      Link to this function
    
    hann(opts \\ [])


      
       
       View Source
     


  


  

Hann window.

  
  options

  
  Options


	:n - The window length. Mandatory option.
	:is_periodic - If true, produces a periodic window,
 otherwise produces a symmetric window. Defaults to true
	:type - the output type for the window. Defaults to {:f, 32}
	:name - the axis name. Defaults to nil


  
  examples

  
  Examples


iex> NxSignal.Windows.hann(n: 5, is_periodic: false)
#Nx.Tensor<
  f32[5]
  [0.0, 0.5, 1.0, 0.5, 0.0]
>
iex> NxSignal.Windows.hann(n: 5, is_periodic: true)
#Nx.Tensor<
  f32[5]
  [0.0, 0.34549152851104736, 0.9045085310935974, 0.9045084714889526, 0.3454914391040802]
>

  



    

  
    
      
      Link to this function
    
    rectangular(opts \\ [])


      
       
       View Source
     


  


  

Rectangular window.
Useful for when no window function should be applied.

  
  options

  
  Options


	:n - the window length
	:type - the output type. Defaults to s64


  
  examples

  
  Examples


iex> NxSignal.Windows.rectangular(n: 5)
#Nx.Tensor<
  s64[5]
  [1, 1, 1, 1, 1]
>

iex> NxSignal.Windows.rectangular(n: 5, type: :f32)
#Nx.Tensor<
  f32[5]
  [1.0, 1.0, 1.0, 1.0, 1.0]
>

  



    

  
    
      
      Link to this function
    
    triangular(opts \\ [])


      
       
       View Source
     


  


  

Triangular window.
See also: bartlett/1

  
  options

  
  Options


	:n - The window length. Mandatory option.
	:type - the output type for the window. Defaults to {:f, 32}
	:name - the axis name. Defaults to nil


  
  examples

  
  Examples


iex> NxSignal.Windows.triangular(n: 3)
#Nx.Tensor<
  f32[3]
  [0.5, 1.0, 0.5]
>

